

```

* Stata Analysis Examples Replication for ASDA 2nd Edition
* Berglund April 2017
* Chapter 9

use "P:\ASDA 2\Data sets\ncsr\ncsr_sub_13nov2015.dta", clear
rename *, lower

* Example 9.2.6 Multinomial logistic regression using NCSR data.
svyset seclustr [pweight = ncsrwtlg], strata(sestrat) vce(linearized) singleunit(missing)

* with coefficients as output Table 9.2
svy: mlogit wkstat3c ib2.sex ald mde i.ed4cat i.ag4cat i.mar3cat

* with risk ratios as output Table 9.3
svy: mlogit wkstat3c ib2.sex ald mde i.ed4cat i.ag4cat i.mar3cat, rrr
* tests Table 9.4
test 2.ag4cat 3.ag4cat 4.ag4cat
test 1.sex
test ald
test mde
test 2.mar3cat 3.mar3cat
test 2.ed4cat 3.ed4cat 4.ed4cat

* Figure 9.3
margins, dydx(ald) by(ag4cat) predict(pr outcome(3))
marginsplot
* Figure 9.4
margins, by(ald ag4cat) predict(pr outcome(3))
marginsplot

* mlogitgof test
svy: mlogit wkstat3c ib2.sex ald mde i.ed4cat i.ag4cat i.mar3cat, rrr
mlogitgof

* 9.3.6 Example: Fitting a Cumulative Logit Regression Model to Complex Sample Survey Data
* Figure 9.6 Russian Fed data.
use "P:\ASDA 2\Data sets\ESS6 Russia\ess6_russia_2aug2016.dta", clear
rename *, lower
tab stflife, missing

* create a variable called stflife2 0-1, 2-4, 5, 6-8, and 9-10
gen stflife2 = 1 if stflife >=0 & stflife <=1
replace stflife2=2 if stflife >=2 & stflife <=4
replace stflife2=3 if stflife ==5
replace stflife2=4 if stflife >=6 & stflife <=8
replace stflife2=5 if stflife >=9 & stflife <=10
tab stflife2

tabulate stflife2, gen(sat)
* Figure 9.6
graph bar sat*[pweight=pspwght], ytitle("Proportions") ///
legend(row(1) lab(1 "0-1") lab(2 "2-4") lab(3 "5") lab(4 "6-8") lab(5 "9-10"))

svyset psu [pweight = pspwght], strata(stratify)

* Table 9.5 and 9.6
svy: ologit stflife2 i.agecat i.marcat male
svy: ologit, or

tabulate agecat, gen(agecat)
tabulate marcat, gen(marcat)

* as of 6apr2017, use gologit29 rather than gologit2
* findit gologit29
gologit29 stflife2 agecat2 agecat3 agecat4 marcat2 marcat3 male, svy autofit

```

```

* 9.4.7 Example: Fitting Poisson and Negative Binomial Regression Models to Complex Sample Survey
Data
use "P:\ASDA 2\Data sets\HRS 2012\hrs_sub_28sep2016.dta", clear
rename *, lower

sum numfalls24 if age65p == 1, detail
sum numfalls24 if numfalls24 >= 1 & age65p == 1, detail

svyset secu [pweight = nwgtr], strata(stratum)

svy, subpop(age65p): mean nage
svy, subpop(age65p): mean r1lbmi

gen nage_c = nage - 74.5
gen bmi_c = r1lbmi - 27.7
gen offset24 = 24

* Table 9.8 and 9.9
svy, subpop(age65p): poisson numfalls24 ib2.gender ///
nage_c arthritis diabetes bmi_c, irr exposure(offset24)

svy, subpop(age65p): nbreg numfalls24 ib2.gender ///
nage_c arthritis diabetes bmi_c, irr exposure(offset24)

svy, subpop(age65p): zinb numfalls24 ib2.gender nage_c ///
arthritis diabetes bmi_c, inflate(ib2.gender nage_c arthritis) irr exposure(offset24)

* obtain OR for Part I of model using manual approach, for example numbers for Male
di exp(1.5107792)
di exp(.5157729)
di exp( 2.505785)

```